Système d'équation linéaires avec LinearAlgebra

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

Ce doument est une révision de celui produit en 2004. L'objectif principal de cette feuille `Maple` est d'initier l'élève à la résolution de systèmes d'équations linéaires. L'extension `LinearAlgebra` dispose de toutes les macro-commandes nécessaires à cette initiation. Dans un premier temps, l'élève apprendra à développer pas à pas la méthode de réduction de Gauss en échelonnant la matrice augmentée d'un système d'équations linéaires à l'aide des opérations élémentaires sur les lignes et à l'aide de la *méthode du pivot.* Dans un deuxième temps, l'élève sera amené à discuter la nature des solutions d'un système d'équations linéaires à l'aide des macro-commandes permettant d'obtenir directement la forme échelonnée ou échelonnée réduite de la matrice augmentée du système à résoudre. Finalement, l'élève verra comment procéder à l'inversion de matrices régulières par la méthode de réduction de Gauss.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> with(LinearAlgebra):
```

Macro-commande LinearAlgebra[GenerateMatrix]

Soit le système d'équations linéaires S suivant:

$$S:\quad \begin{aligned} x_1 - 2x_2 + x_3 &= -1 \\ x_1 + x_2 - x_3 &= 2 \\ 3x_1 + 6x_2 - 4x_3 &= 1 \end{aligned}$$

```
> S:=[  x[1] - 2*x[2]  +   x[3]  = -1 ,
        x[1] +   x[2]  -   x[3]  =  2 ,
      3*x[1] + 6*x[2]  - 4*x[3]  =  1 ];
```

$$S := [x_1 - 2x_2 + x_3 = -1, x_1 + x_2 - x_3 = 2, 3x_1 + 6x_2 - 4x_3 = 1] \quad \textbf{(2.1)}$$

La macro-commande GenerateMatrix de la bibilothèque `LinearAlgebra` permet de créer, à la fois, la matrice des coefficients et la matrice colonne (vecteur colonne) des termes constants d'un système d'équations linéaires:

– le premier argument de la macro-commande doit être la séquence des équations formant le système. Cette séquence qui doit être formulée comme un objet de type **`set`** ou de type **`list`**;

– le deuxième argument de la macro-commande doit être la liste des inconnues du système;

– le troisième argument, « `augmented=true` » est optionnel: cette option permet de créer directement la matrice augmentée du système d'équations linéaires.

Attention:

- Si on omet le troisième argument, le résultat de `GenerateMatrix` est une séquence de deux objets: soit la matrice des coefficients du système d'équations linéaires ainsi que la matrice colonne des termes constants.
- Si on précise ce troisième argument, le résultat de `GenerateMatrix` est exactement la matrice augmentée du système d'équations linéaires. Le résultat n'est donc pas une séquence d'objets.

```
> A,b:=GenerateMatrix(S,[seq(x[i],i=1..3)]);
```

$$A, b := \begin{bmatrix} 1 & -2 & 1 \\ 1 & 1 & -1 \\ 3 & 6 & -4 \end{bmatrix}, \begin{bmatrix} -1 \\ 2 \\ 1 \end{bmatrix} \tag{2.2}$$

```
> whattype(A);
```

$$Matrix \tag{2.3}$$

```
> whattype(b);
```

$$Vector_{column} \tag{2.4}$$

```
> `[A|b]`:=GenerateMatrix(S,[seq(x[i],i=1..3)],augmented=true);
```

$$[A|b] := \begin{bmatrix} 1 & -2 & 1 & -1 \\ 1 & 1 & -1 & 2 \\ 3 & 6 & -4 & 1 \end{bmatrix} \tag{2.5}$$

On peut très bien obtenir la matrice augmentée à partir de la matrice A et du vecteur colonne b obtenus précédemment. Utilisons le raccourci ligne <|>.

```
> `[A|b]`:=<A|b>;
```

$$[A|b] := \begin{bmatrix} 1 & -2 & 1 & -1 \\ 1 & 1 & -1 & 2 \\ 3 & 6 & -4 & 1 \end{bmatrix} \tag{2.6}$$

```
> whattype(`[A|b]`);
```

$$Matrix \tag{2.7}$$

Méthode de réduction de Gauss

Opérations élémentaires de lignes avec la macro-commande LinearAlgebra [RowOperation]

Voici la syntaxe `Maple` des opérations élémentaires sur les lignes d'une matrice M.

(a) permutation de la *i-ième* ligne et de la *j-ième* ligne .... $RowOperation(M, [i, j])$

(b) multiplication de la *i-ième* ligne par un scalaire k .... $RowOperation(M, i, k)$

(c) addition à la *i-ième ligne* d'un multiple k de la *j-ième* ligne .... $RowOperation(M, [i, j], k)$

Effectuons l'échelonnage de la matrice augmentée $[A \mid b]$ du système d'équations linéaires S.

Rappelons-nous le détail de la matrice `[A | b]`.

```
> '`[A|b]`'=`[A|b]`;
```

$$[A|b] = \begin{bmatrix} 1 & -2 & 1 & -1 \\ 1 & 1 & -1 & 2 \\ 3 & 6 & -4 & 1 \end{bmatrix} \quad \textbf{(3.1.1)}$$

Additionnons à la deuxième ligne, (-1) fois la première ligne.

```
> M1:=RowOperation(`[A|b]`,[2,1],-1);
```

$$M1 := \begin{bmatrix} 1 & -2 & 1 & -1 \\ 0 & 3 & -2 & 3 \\ 3 & 6 & -4 & 1 \end{bmatrix} \quad \textbf{(3.1.2)}$$

Additionnons à la troisième ligne, (-3) fois la première ligne.

```
> M2:=RowOperation(M1,[3,1],-3);
```

$$M2 := \begin{bmatrix} 1 & -2 & 1 & -1 \\ 0 & 3 & -2 & 3 \\ 0 & 12 & -7 & 4 \end{bmatrix} \quad \textbf{(3.1.3)}$$

Additionnons à la troisième ligne, (-4) fois la deuxième ligne

```
> M3:=RowOperation(M2,[3,2],-4);
```

$$M3 := \begin{bmatrix} 1 & -2 & 1 & -1 \\ 0 & 3 & -2 & 3 \\ 0 & 0 & 1 & -8 \end{bmatrix} \quad \textbf{(3.1.4)}$$

La forme de la matrice M3 est appelée ***matrice échelonnée*** *(selon les lignes)*:

- le premier élément non nul de chaque ligne est situé à droite du premier élément non nul de la ligne précédente. Cet élément est appelé le pivot de la ligne.
  Dans certains manuels, on exige que ce premier élément non nul soit l'unité, c'est-à-dire que chaque pivot doit être égale à l'unité.
- toutes les lignes contenant que des zéros (s'il y a de telles lignes) sont situées sous les lignes qui contiennent des éléments non nuls.

En associant la matrice M3 au système $S3$ ci-dessous,

$$S3 : \quad \begin{array}{rcrcrcr} x_1 & - & 2x_2 & + & x_3 & = & -1 \\ & & 3x_2 & - & 2x_3 & = & 3 \\ & & & & x_3 & = & -8 \end{array}$$

on résout facilement, avec un calcul *manu scriptus*, le système $S3$ par substitution à rebours. Pour ce faire, à partir de la troisième équation, on déduirait que la solution du système d'équations linéaires $S3$ est $x_3 = -8$, ensuite que $x_2 = -\frac{13}{3}$ et finalement que $x_1 = -\frac{5}{3}$.

Par l'application des opérations élémentaires sur les lignes, on a que le système $S3 \sim S$. Ce qui montre que le système d'équations linéaires original S est un système possédant une solution unique.

La macro-commande BackwardSubstitute de la bibliothèque `LinearAlgebra` automatise la substitution à rebours lorsqu'on l'applique à la *matrice échelonnée* du système d'équations S.

```
> Solution=BackwardSubstitute(M3);
```

$$Solution = \begin{bmatrix} -\frac{5}{3} \\ -\frac{13}{3} \\ -8 \end{bmatrix} \quad \textbf{(3.1.5)}$$

```
> whattype(rhs(%));
```

$$Vector_{column} \quad \textbf{(3.1.6)}$$

Le résultat un objet de type Vector dont l'ordre des composantes correspond à l'ordre des inconnues.

Formulons la réponse comme suit dans la zone de texte ci-dessous:

Réponse: $x_1 = -\frac{5}{3}$, $x_2 = -\frac{13}{3}$ et $x_3 = -8$ ou E.-S.= $\left\{ \left[-\frac{5}{3}, -\frac{13}{3}, -8 \right]^t \right\}$.

Poursuivons avec les opérations élémentaires sur les lignes afin de rendre unitaire chaque pivot de la matrice échelonnée.

Pour déduire la forme échelonnée de la matrice A, il suffit, ici, de multiplier la deuxième ligne de la matrice M3 par 1/3.

```
> M4:=RowOperation(M3,2,1/3);
```

$$M4 := \begin{bmatrix} 1 & -2 & 1 & -1 \\ 0 & 1 & -\frac{2}{3} & 1 \\ 0 & 0 & 1 & -8 \end{bmatrix} \quad \textbf{(3.1.7)}$$

Pour obtenir le vecteur colonne solution du système, on peut, également bien sûr, appliquer la macro-commande BackwardSubstitute sur la matrice échelonnée précédente.

```
> Solution=BackwardSubstitute(M4);
```

$$Solution = \begin{bmatrix} -\frac{5}{3} \\ -\frac{13}{3} \\ -8 \end{bmatrix} \quad \textbf{(3.1.8)}$$

On a formulé précédemment, dans une zone de texte, le système d'équations linéaires correspondant à la matrice échelonnée M3. Nous allons montré qu'il est possible aussi de le faire directement dans une zone de résultats.

La macro-commande GenerateEquations de la bibilothèque `LinearAlgebra` permet de créer un système d'équations linéaires à partir d'une matrice désignée comme matrice des coefficients ou comme matrice augmentée d'un système:

- le premier argument de la macro-commande doit être la matrice considérée comme la matrice des coefficients du système d'équations linéaires qu'on veut créer
- le deuxième argument de la macro-commande doit être la liste des inconnues du système. Si le deuxième argument est seulement un nom de variable (pas un objet de type **`list`**), alors c'est la notation indicielle correspondant à cette variable qui sera pris en compte pour les variables du système
- si le troisième argument, qui est optionnel, est un vecteur, alors ce vecteur sera pris en compte comme le vecteur colonne des termes constants du système qui sera créer. Si on omet ce troisième argument, c'est le vecteur nul qui sera pris en compte, ce qui générera un système homogène.

Comme exemple, générons le système d'équations associé à la matrice M4. Comme cette matrice n'est pas la matrice des coefficients mais plutôt la matrice augmentée du système S, il faut d'abord en extraire la matrice des coefficients. Utilisons SubMatrix de l'extension `LinearAlgebra`.

Rappelons-nous d'abord la matrice M4;

```
> M4;
```

$$\begin{bmatrix} 1 & -2 & 1 & -1 \\ 0 & 1 & -\frac{2}{3} & 1 \\ 0 & 0 & 1 & -8 \end{bmatrix} \quad \textbf{(3.1.9)}$$

```
> A_échelon:=SubMatrix(M4,1..3,1..3);
```

$$A\_échelon := \begin{bmatrix} 1 & -2 & 1 \\ 0 & 1 & -\frac{2}{3} \\ 0 & 0 & 1 \end{bmatrix} \quad \textbf{(3.1.10)}$$

Extrayons ensuite la quatrième colonne de la matrice M4.

```
> b_échelon:=SubMatrix(M4,1..3,4..4);
```

$$b\_échelon := \begin{bmatrix} -1 \\ 1 \\ -8 \end{bmatrix} \quad \textbf{(3.1.11)}$$

Créons maintenant le système d'équations linéaires associé à la matrice augmentée M4

```
> `S'`:=GenerateEquations(A_échelon,[seq(x[i],i=1..3)],b);
```

$$S' := \left[x_1 - 2x_2 + x_3 = -1, x_2 - \frac{2x_3}{3} = 2, x_3 = 1\right] \quad \textbf{(3.1.12)}$$

Soit donc le système $x_1 - 2x_2 + x_3 = -1$

$$x_2 - \frac{2x_3}{3} = 2$$

$$x_3 = 1$$

Notons que le système d'équations (le résultat de la macro-commande) est formulé en tant qu'objet de type **`list`**.

```
> whattype(`S'`);
```

$$list \qquad \textbf{(3.1.13)}$$

Macro-commande LinearAlgebra[GaussianElimination]

La macro-commande GaussianElimination permet d'obtenir automatiquement la *matrice échelon* (le pivot pas nécessairement unitaire) de la matrice augmentée du système d'équations linéaires.

```
> G:=GaussianElimination(`[A|b]`);
```

$$G := \begin{bmatrix} 1 & -2 & 1 & -1 \\ 0 & 3 & -2 & 3 \\ 0 & 0 & 1 & -8 \end{bmatrix} \qquad \textbf{(3.2.1)}$$

Méthode de réduction de Gauss-Jordan

Opérations élémentaires de lignes (suite)

Les opérations élémentaires de lignes peuvent être utilisées pour transformer davantage la matrice échelonnée. Poursuivons les transformations de la matrice *[A|b]* à partir de la matrice échelonnée M3 de telle sorte que le pivot de chaque ligne soit l'unique élément non nul de sa colonne.

Rappelons-nous la matrice échelonnée M3

```
> M3;
```

$$\begin{bmatrix} 1 & -2 & 1 & -1 \\ 0 & 3 & -2 & 3 \\ 0 & 0 & 1 & -8 \end{bmatrix} \qquad \textbf{(4.1.1)}$$

Additionnons à la deuxième ligne, 2 fois la troisième ligne.

```
> M4:=RowOperation(M3,[2,3],2);
```

$$M4 := \begin{bmatrix} 1 & -2 & 1 & -1 \\ 0 & 3 & 0 & -13 \\ 0 & 0 & 1 & -8 \end{bmatrix} \qquad \textbf{(4.1.2)}$$

Additionnons à la première ligne, (-1) fois la troisième ligne.

```
> M5:=RowOperation(M4,[1,3],-1);
```

$$M5 := \begin{bmatrix} 1 & -2 & 0 & 7 \\ 0 & 3 & 0 & -13 \\ 0 & 0 & 1 & -8 \end{bmatrix} \qquad \textbf{(4.1.3)}$$

Additionnons à la première ligne,

$\frac{2}{3}$ fois la deuxième ligne.

```
> M6:=RowOperation(M5,[1,2],2/3);
```

$$M6 := \begin{bmatrix} 1 & 0 & 0 & -\frac{5}{3} \\ 0 & 3 & 0 & -13 \\ 0 & 0 & 1 & -8 \end{bmatrix} \quad \textbf{(4.1.4)}$$

Poursuivons les transformations afin de rendre le premier élément non nul de chaque ligne égale à l'unité. Ici, il suffit seulement de multiplier la deuxième ligne de la matrice M6 par $\frac{1}{3}$.

```
> M7:=RowOperation(M6,2,1/3);
```

$$M7 := \begin{bmatrix} 1 & 0 & 0 & -\frac{5}{3} \\ 0 & 1 & 0 & -\frac{13}{3} \\ 0 & 0 & 1 & -8 \end{bmatrix} \quad \textbf{(4.1.5)}$$

La matrice M7 satisfait la définition d'une matrice échelonnée réduite:

– le pivot de chaque ligne est égal à l'unité

– le pivot d'une ligne est toujours à la droite du pivot de la ligne précédente

– dans la colonne du pivot de chaque ligne, il est le seul élément non nul de la colonne

– s'il y a des lignes dont tous les éléments sont nuls, elles composent les dernières lignes de la matrice.

Technique du pivotage

La matrice échelonnée réduite M7 a été obtenue en utilisant les opérations élémentaires de lignes sur la matrice augmentée $[A|b]$ du système d'équations linéaires S. Nous allons maintenant obtenir de nouveau la *matrice échelonnée réduite* du système S mais, cette fois, par la *technique du pivotage*.

Rappelons-nous la matrice augmentée du système S.

```
> M:=`[A|b]`;
```

$$M := \begin{bmatrix} 1 & -2 & 1 & -1 \\ 1 & 1 & -1 & 2 \\ 3 & 6 & -4 & 1 \end{bmatrix} \quad \textbf{(4.2.1)}$$

La technique du pivotage consiste à

– désigner comme pivot un élément non nul m_{ij} d'une matrice M.

  – la *i-ème* ligne est alors appelée ligne pivot

  – la *j-ème* colonne est alors appelée colonne pivot

– additionner à chacune des autres lignes un multiple approprié de la ligne pivot afin d'annuler toutes les autres entrées de la colonne pivot

En désignant l'entrée m_{ij} pour effectuer le pivotage, la macro-commande Pivot de la bibliothèque `LinearAlgebra` annule automatiquement toutes les entrées de la *j-ième* colonne sauf évidemment l'élément m_{ij} qui a été désigné comme pivot. La technique du pivotage est en quelque sorte une façon d'effectuer plusieurs opérations élémentaires `RowOperation` dans un ordre différent.

Désignons l'élément m_{11} de la matrice M comme pivot dans la macro-commande **`pivot`**.

```
> M1:=Pivot(M,1,1);
```

$$M1 := \begin{bmatrix} 1 & -2 & 1 & -1 \\ 0 & 3 & -2 & 3 \\ 0 & 12 & -7 & 4 \end{bmatrix} \qquad \textbf{(4.2.2)}$$

Désignons l'élément $m1_{22}$ comme pivot.

```
> M2:=Pivot(M1,2,2);
```

$$M2 := \begin{bmatrix} 1 & 0 & -\frac{1}{3} & 1 \\ 0 & 3 & -2 & 3 \\ 0 & 0 & 1 & -8 \end{bmatrix} \qquad \textbf{(4.2.3)}$$

Désignons l'élément $m2_{33}$ comme pivot.

```
> M3:=Pivot(M2,3,3);
```

$$M3 := \begin{bmatrix} 1 & 0 & 0 & -\frac{5}{3} \\ 0 & 3 & 0 & -13 \\ 0 & 0 & 1 & -8 \end{bmatrix} \qquad \textbf{(4.2.4)}$$

Pour obtenir la matrice échelonnée réduite du système S, il ne reste qu'à rendre unitaire l'élément $m3_{22}$

```
> M4:=RowOperation(M3,2,1/3);
```

$$M4 := \begin{bmatrix} 1 & 0 & 0 & -\frac{5}{3} \\ 0 & 1 & 0 & -\frac{13}{3} \\ 0 & 0 & 1 & -8 \end{bmatrix} \qquad \textbf{(4.2.5)}$$

La technique du pivotage a permis d'obtenir la matrice échelonnée réduite plus rapidement que par l'application seules des opérations élémentaires de lignes.

Remarque: Dans le cas où il suffit seulement de résoudre un système d'équations linéaires par la technique du pivotage, il n'est pas nécessaire de désigner, dans l'ordre, les éléments m_{11}, m_{22}, m_{33}, etc., comme pivot. L'exemple ci-dessous va le montrer.

Soit le système T suivant à résoudre.

$$T:\ \begin{cases} 3x_1 - 2x_2 - 3x_3 = 4 \\ 3x_1 + 2x_2 + x_3 = 0 \\ 2x_1 + x_2 + x_3 = 1 \end{cases}$$

```
> T:=[  3*x[1] - 2*x[2] - 3*x[3] = 4 ,
        3*x[1] + 2*x[2] +   x[3] = 0 ,
        2*x[1] +   x[2] +   x[3] = 1 ];
```

$$T := [3x_1 - 2x_2 - 3x_3 = 4, 3x_1 + 2x_2 + x_3 = 0, 2x_1 + x_2 + x_3 = 1] \quad \textbf{(4.2.6)}$$

```
> M:=GenerateMatrix(T,[seq(x[i],i=1..3)],augmented=true);
```

$$M := \begin{bmatrix} 3 & -2 & -3 & 4 \\ 3 & 2 & 1 & 0 \\ 2 & 1 & 1 & 1 \end{bmatrix} \quad \textbf{(4.2.7)}$$

La réalisation du pivotage peut très bien se faire dans un ordre différent.
Désignons en premier l'élément m_{32} comme pivot.

```
> M1:=Pivot(M,3,2);
```

$$M1 := \begin{bmatrix} 7 & 0 & -1 & 6 \\ -1 & 0 & -1 & -2 \\ 2 & 1 & 1 & 1 \end{bmatrix} \quad \textbf{(4.2.8)}$$

Désignons l'élément $m1_{23}$ comme pivot.

```
> M2:=Pivot(M1,2,3);
```

$$M2 := \begin{bmatrix} 8 & 0 & 0 & 8 \\ -1 & 0 & -1 & -2 \\ 1 & 1 & 0 & -1 \end{bmatrix} \quad \textbf{(4.2.9)}$$

Désignons l'élément $m2_{11}$ comme pivot.

```
> M3:=Pivot(M2,1,1);
```

$$M3 := \begin{bmatrix} 8 & 0 & 0 & 8 \\ 0 & 0 & -1 & -1 \\ 0 & 1 & 0 & -2 \end{bmatrix} \quad \textbf{(4.2.10)}$$

Multiplions la première ligne de la matrice M3 par $\frac{1}{8}$.

```
> M4:=RowOperation(M3,1,1/8);
```

$$M4 := \begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 0 & -1 & -1 \\ 0 & 1 & 0 & -2 \end{bmatrix} \quad \textbf{(4.2.11)}$$

Multiplions la deuxième ligne de la matrice M4 par -1.

```
> M5:=RowOperation(M4,2,-1);
```

(4.2.12)

$$M5 := \begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 0 & 1 & 1 \\ 0 & 1 & 0 & -2 \end{bmatrix} \tag{4.2.12}$$

Même si le dernier résultat ne correspond pas à la forme échelonnée réduite de la matrice augmentée, on peut quand même déduire l'ensemble-solution du système T: $x_1 = 1$, $x_2 = -2$ et $x_3 = 1$ puisque la matrice augmentée correspond au système équivalent ci -dessous.

$$\begin{array}{lcl} x_1 & = & 1 \\ \quad\quad x_3 & = & 1 \\ \quad x_2 & = & -2 \end{array}$$

Macro-commandes LinearAlgebra[GaussianElimination] et LinearAlgebra [ReducedRowEchelonForm]

La macro-commande GaussianElimination de l'extension `LinearAlgebra` donne comme résultat une matrice échelonnée. Appliquons cette macro-commande à la matrice augmentée M du système T.

```
> M_échelonnée:=GaussianElimination(M);
```

$$M\_échelonnée := \begin{bmatrix} 3 & -2 & -3 & 4 \\ 0 & 4 & 4 & -4 \\ 0 & 0 & \frac{2}{3} & \frac{2}{3} \end{bmatrix} \tag{4.3.1}$$

Il est possible d'obtenir une matrice échelon sont introduire de fraction.

```
> M_échelonnée:=GaussianElimination(M,'method'='FractionFree');
```

$$M\_échelonnée := \begin{bmatrix} 3 & -2 & -3 & 4 \\ 0 & 12 & 12 & -12 \\ 0 & 0 & 8 & 8 \end{bmatrix} \tag{4.3.2}$$

La macro-commande ReducedRowEchelonForm l'extension `LinearAlgebra` donne directement pour résultat la forme échelonnée réduite. Appliquons cette macro-commande à la matrice augmentée M du système T.

```
> M_échelonnée_réduite:=ReducedRowEchelonForm(M);
```

$$M\_échelonnée\_réduite := \begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & -2 \\ 0 & 0 & 1 & 1 \end{bmatrix} \tag{4.3.3}$$

Nature des solutions d'un système d'équations linéaires

Système contradictoire

Résolvons l'équation matricielle S: $\begin{bmatrix} 2 & 3 & -1 \\ 1 & 1 & 2 \\ 1 & 2 & -3 \end{bmatrix} \begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} -1 \\ 12 \\ 2 \end{bmatrix}$.

Puisque le système à résoudre est présenté avec l'écriture matricielle AX= b, ce que l'on recherche est alors une matrice colonne X satisfaisant cette équation matricielle.

Transposons en `Maple` l'écriture matricielle du système d'équations linéaires S.

Créons la matrice des coefficients du système *S*.

```
> A:=Matrix([
      [2,3,-1],
      [1,1,2],
      [1,2,-3] ]);
```

$$A := \begin{bmatrix} 2 & 3 & -1 \\ 1 & 1 & 2 \\ 1 & 2 & -3 \end{bmatrix} \quad \textbf{(5.1.1)}$$

Créons la matrice colonne des inconnues du système *S*.

```
> X:=Vector([x,y,z]);
```

$$X := \begin{bmatrix} x \\ y \\ z \end{bmatrix} \quad \textbf{(5.1.2)}$$

C'est plus concis en créant la matrice colonne des inconnues avec le raccourci colonne <,>.

```
> X:=<z,y,z>;
```

$$X := \begin{bmatrix} z \\ y \\ z \end{bmatrix} \quad \textbf{(5.1.3)}$$

Créons la matrice colonne des termes constants du système *S*.

```
> b:=<-1,12,2>;
```

$$b := \begin{bmatrix} -1 \\ 12 \\ 2 \end{bmatrix} \quad \textbf{(5.1.4)}$$

Ainsi, on pourra donc afficher le système *S* avec une écriture matricielle dans une zone des résultats.

```
> A.X = b;
```

$$\begin{bmatrix} z+3y \\ 3z+y \\ -2z+2y \end{bmatrix} = \begin{bmatrix} -1 \\ 12 \\ 2 \end{bmatrix} \quad \textbf{(5.1.5)}$$

Pour ensuite créer la matrice augmentée associée au système d'équations avec le racourci ligne <|>.

```
> `A|b`:=<A|b>;
```

$$A|b := \begin{bmatrix} 2 & 3 & -1 & -1 \\ 1 & 1 & 2 & 12 \\ 1 & 2 & -3 & 2 \end{bmatrix} \quad \textbf{(5.1.6)}$$

Pour déterminer la nature du système d'équations, obtenons directement la forme de Gauss-Jordan de la matrice $A|B$ à l'aide de la macro-commande **rref**.

```
> `A|b_échelon_réduite`:=ReducedRowEchelonForm(`A|b`);
```

$$A|b\_échelon\_réduite := \begin{bmatrix} 1 & 0 & 7 & 0 \\ 0 & 1 & -5 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad \textbf{(5.1.7)}$$

La dernière ligne de la matrice échelonnée réduite nous montre clairement que le système à résoudre est un système incohérent. Dans une zone de texte, reste à formuler la réponse.
Réponse: E-S = { } (ou = ∅)

Par curiosité, appliquons quand même la macro-commande `BackwardSubstitute` sur cette matrice pour "voir" comment l'évaluateur traitera la requête.

```
> X=BackwardSubstitute(`A|b_échelon_réduite`);
Error, (in LinearAlgebra:-BackwardSubstitute) inconsistent system
```

Système cohérent avec

solution unique

Résolvons le système S: $\begin{bmatrix} 1 & -2 & 3 \\ 2 & 2 & -1 \\ 1 & 1 & 5 \\ 1 & 1 & -6 \end{bmatrix} \begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} 16 \\ 0 \\ 11 \\ -11 \end{bmatrix}$.

Créons la matrice des coefficients du système *S*.

```
> A:=Matrix([
    [1, -2, 3],
    [2, 2, -1],
    [1, 1, 5],
    [1, 1, -6] ]);
```

$$A := \begin{bmatrix} 1 & -2 & 3 \\ 2 & 2 & -1 \\ 1 & 1 & 5 \\ 1 & 1 & -6 \end{bmatrix} \quad \textbf{(5.2.1.1)}$$

Créons la matrice colonne des termes constants du système *S*.

```
> b:=<16,0,11,-11>;
```

$$b := \begin{bmatrix} 16 \\ 0 \\ 11 \\ -11 \end{bmatrix} \qquad \textbf{(5.2.1.2)}$$

Créons la matrice augmentée du système *S*.

```
> `A|b`:=<A|b>;
```

$$A|b := \begin{bmatrix} 1 & -2 & 3 & 16 \\ 2 & 2 & -1 & 0 \\ 1 & 1 & 5 & 11 \\ 1 & 1 & -6 & -11 \end{bmatrix} \qquad \textbf{(5.2.1.3)}$$

Obtenons directement avec la macro-commande `ReducedRowEchelonForm` la forme de Gauss-Jordan de la matrice A|B.

```
> `A|b_échelon_réduite`:=ReducedRowEchelonForm(`A|b`);
```

$$A|b\_échelon\_réduite := \begin{bmatrix} 1 & 0 & 0 & 4 \\ 0 & 1 & 0 & -3 \\ 0 & 0 & 1 & 2 \\ 0 & 0 & 0 & 0 \end{bmatrix} \qquad \textbf{(5.2.1.4)}$$

La matrice $A|B\_échelon\_réduite$ associée au système *S* nous montre que le système est un système cohérent avec solution unique. La matrice de Gauss-Jordan nous donne, sans calcul à faire, la solution du système S.

À cause de la notation matricielle du système S, la solution doit être formulée comme un vecteur colonne.

Réponse: E-S $= \left\{ \begin{bmatrix} 4 \\ -3 \\ 2 \end{bmatrix} \right\}$.

Mais, il est plus commode d'utiliser la notation transposée.
Réponse: E-S = $\{ [4, -3, 2]^t \}$

Utilisons quand même la macro-commande `BackwardSubstitute` et observons le résultat qui s'affichera.

```
> X=BackwardSubstitute(`A|b_échelon_réduite`);
```

$$\begin{bmatrix} z \\ y \\ z \end{bmatrix} = \begin{bmatrix} 4 \\ -3 \\ 2 \end{bmatrix} \qquad \textbf{(5.2.1.5)}$$

une infinité de solutions

Résolvons le système S: $\begin{bmatrix} 0 & -3 & -6 & 4 \\ -1 & -2 & -1 & 3 \\ -2 & -3 & 0 & 3 \\ 1 & 4 & 5 & -9 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{bmatrix} = \begin{bmatrix} 9 \\ 1 \\ -1 \\ -7 \end{bmatrix}$.

Obtenons la matrice augmentée du système *S*.

```
> A:=Matrix([
      [ 0,-3,-6, 4],
      [-1,-2,-1, 3],
      [-2,-3, 0, 3],
      [ 1, 4, 5,-9]  ]):
  b:=<9,1,-1,-7>:
  `A|b`:=<A|b>;
```

$$A|b := \begin{bmatrix} 0 & -3 & -6 & 4 & 9 \\ -1 & -2 & -1 & 3 & 1 \\ -2 & -3 & 0 & 3 & -1 \\ 1 & 4 & 5 & -9 & -7 \end{bmatrix} \tag{5.2.2.1}$$

Contruisons le vecteur colonne des inconnues.

```
> X:=Vector([seq(x[i],i=1..4)]);
```

$$X := \begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{bmatrix} \tag{5.2.2.2}$$

Obtenons directement la matrice de Gauss-Jordan de A|B

```
> `A|b_échelon_réduite`:=ReducedRowEchelonForm(`A|b`);
```

$$A|b\_échelon\_réduite := \begin{bmatrix} 1 & 0 & -3 & 0 & 5 \\ 0 & 1 & 2 & 0 & -3 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix} \tag{5.2.2.3}$$

La matrice *A|B_échelon_réduite* associée au système à résoudre nous indique que le système *S* possède une infinité de solutions. Utilisons la macro-commande **backsub** et observons, dans ce cas, la formulation du résultat.

```
> X=BackwardSubstitute(`A|b_échelon_réduite`);
```

(5.2.2.4)

$$\begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{bmatrix} = \begin{bmatrix} 5+3\_t3_1 \\ -3-2\_t3_1 \\ \_t3_1 \\ 0 \end{bmatrix} \quad \textbf{(5.2.2.4)}$$

L'utilisation du trait souligné nous indique que $\_t_1$ est une variable mathématique qui a été automatiquement générée par le simplificateur. Dans le contexte actuel, on doit interpréter $\_t_1$ comme un paramètre. On constate donc que le système possède une infinité de solutions et que la formulation donnée du résultat est une formulation paramétrique où l'évaluateur a désigné la variable x_3 comme étant la variable libre du système.

Dans une zone de texte alors, formulons donc l'ensemble-solution avec la notation habituelle d'un l'ensemble-solution dans le cas d'un système indéterminé.
Réponse: E-S = $\{ [5+3s, -3-2s, s, 0]^t \mid s \in \mathbb{R}\}$.

Dans un certain contexte (à la demande de votre professeur par exemple…), il peut être plus intéressant de reformuler, dans une zone de résultats, *Solution* avec une notation mathématique plus habituelle. Il suffira, dans ce cas, de faire une substitution de noms.

```
> subs(_t10[1]=t,(5.2.2.4));
```

$$\begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{bmatrix} = \begin{bmatrix} 5+3\_t3_1 \\ -3-2\_t3_1 \\ \_t3_1 \\ 0 \end{bmatrix} \quad \textbf{(5.2.2.5)}$$

Macro-commande LinearAlgebra[LinearSolve]

Résolvons de nouveau le système S: $\begin{bmatrix} 0 & -3 & -6 & 4 \\ -1 & -2 & -1 & 3 \\ -2 & -3 & 0 & 3 \\ 1 & 4 & 5 & -9 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{bmatrix} = \begin{bmatrix} 9 \\ 1 \\ -1 \\ -7 \end{bmatrix}$ mais, cette fois, en utilisant la macro-commande LinearSolve de la bibliothèque `LinearAlgebra`.

Les deux premiers arguments obligatoires de la macro-commande **`linsolve`** sont, dans l'ordre,

– soit la matrice des coefficients et la matrice colonne des termes constants

– soit la matrices des coefficients et le vecteur (colonne) des termes constants.

Créons de nouveau la matrice A des coefficients et le vecteur colonne b des termes constants.

```
> A:=<< 0,-1,-2, 1>|
      <-3,-2,-3, 4>|
```

```
     <-6,-1, 0, 5>|
     < 4, 3, 3,-9>>;
  b:=<9,1,-1,-7>;
```

$$A := \begin{bmatrix} 0 & -3 & -6 & 4 \\ -1 & -2 & -1 & 3 \\ -2 & -3 & 0 & 3 \\ 1 & 4 & 5 & -9 \end{bmatrix}$$

$$b := \begin{bmatrix} 9 \\ 1 \\ -1 \\ -7 \end{bmatrix} \quad \textbf{(6.1)}$$

```
>
```

Résolvons le système avec la macro-commande `LinearSolve` en précisant l'option `free='t'` afin de désigner le symbole à utiliser pour la ou les variables libres .

```
> X=LinearSolve(A,b,free='t');
```

$$\begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{bmatrix} = \begin{bmatrix} 5+3\,t_3 \\ -3-2\,t_3 \\ t_3 \\ 0 \end{bmatrix} \quad \textbf{(6.2)}$$

Inversion de matrices par la méthode de réduction de Gauss

Trouvons la matrice inverse, si elle existe, de la matrice $A = \begin{bmatrix} 3 & 2 & 4 \\ 2 & -1 & 2 \\ -1 & 1 & 2 \end{bmatrix}$ en utilisant la méthode de réduction de Gauss.

Comme on l'a vu en classe, commençons d'abord par augmenter la matrice A avec la matrice identité.

Créons la matrice A.

```
> A:=< <3,2,-1>|
       <2,-1,1>|
       <4,2,2> >;
```

$$A := \begin{bmatrix} 3 & 2 & 4 \\ 2 & -1 & 2 \\ -1 & 1 & 2 \end{bmatrix} \quad \textbf{(7.1)}$$

```
> `A|I`:=<A|IdentityMatrix(3)>;
```

$$A|I := \begin{bmatrix} 3 & 2 & 4 & 1 & 0 & 0 \\ 2 & -1 & 2 & 0 & 1 & 0 \\ -1 & 1 & 2 & 0 & 0 & 1 \end{bmatrix} \tag{7.2}$$

Reste maintenant à obtenir directement la forme de Gauss-Jordan de la matrice A|Id avec la macro-commande `ReducedRowEchelonForm`.

```
> B:=ReducedRowEchelonForm(`A|I`);
```

$$B := \begin{bmatrix} 1 & 0 & 0 & \frac{1}{5} & 0 & -\frac{2}{5} \\ 0 & 1 & 0 & \frac{3}{10} & -\frac{1}{2} & -\frac{1}{10} \\ 0 & 0 & 1 & -\frac{1}{20} & \frac{1}{4} & \frac{7}{20} \end{bmatrix} \tag{7.3}$$

Le résultat obtenu montre clairement que la matrice A est régulière puisque la matrice B est de la forme I | A^{-1}.

Pour formuler la réponse dans une zone de résultats, utilisons la macro-commande SubMatrix pour extraire la matrice inverse de la matrice B.

```
> A_inverse:=SubMatrix(B,1..3,4..6);
```

$$A\_inverse := \begin{bmatrix} \frac{1}{5} & 0 & -\frac{2}{5} \\ \frac{3}{10} & -\frac{1}{2} & -\frac{1}{10} \\ -\frac{1}{20} & \frac{1}{4} & \frac{7}{20} \end{bmatrix} \tag{7.4}$$

Vérification:

```
> 'A.A_inverse' = A.A_inverse;
```

$$A \cdot A\_inverse = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \tag{7.5}$$

```
> 'A_inverse.A' = A_inverse.A;
```

$$A\_inverse \cdot A = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \tag{7.6}$$